

**Tutorium zu TILO****Sprache einer Grammatik****SoSe 24****16.05.2024****Aufgabe: (Sprache einer Grammatik als Prolog-Programm)**

Zur Überprüfung, ob Ihre Grammatiken wirklich das machen, was sie machen sollen, setzen wir die formalen Definitionen der Reduktionsrelation und der von einer Grammatik erzeugten Sprache in ein Prolog-Programm um.

- Stellen Sie eine beliebige Grammatik als Prolog-Programm dar, indem Sie Relationen für die vier Komponenten einer Grammatik angeben. Dieser Teil ist grammatikabhängig, d.h. bei Verwendung einer anderen Grammatik abzuändern.  
Hierbei sind nur Fakten und keine Prolog-Variablen erlaubt.
- Implementieren Sie eine Relation `redrel(Alphas,Betas)`, die allgemein der Reduktionsrelation für die in a) definierte Grammatik entspricht, indem Sie die formale Definition in Prolog umsetzen.
- Implementieren Sie eine Relation `lvong(Ws)`, die allgemein der Sprache entspricht, die von der in a) definierten Grammatik erzeugt wird.  
Hierzu wird die Implementierung der Relationen
  - `redrel_plus(Alphas,Betas)` für den transitiven Abschluss von `redrel(Alphas,Betas)` und
  - `sigma_stern(Ws)` der Menge der Wörter über Sigma benötigt.

Die Teile b) und c) sind allgemein und nur in Abhängigkeit von den in a) definierten Relationen zu implementieren.

D.h., dass Sie bei Verwendung einer anderen Grammatik nur die Teile aus a) abändern müssen.

**Setzen Sie die Formalismen der Vorlesung 1 zu 1 in Prolog um!!!****Lösung:**

```
%spezielle Grammatik für L_Palin-Bin:
%Variablenmenge:
variablen(vS).
%Alphabet:
sigma(0).
sigma(1).
%Startvariable:
start(vS).
%Regeln:
regeln([vS],[ ]).
regeln([vS],[0]).
regeln([vS],[1]).
regeln([vS],[0,vS,0]).
regeln([vS],[1,vS,1]).
%allgemeiner Teil:
redrel(Alphas,Betas) :- regeln(Ls,Rs),
                        append(Alpha1s,Ls,H1s),
                        append(H1s,Alpha2s,Alphas),
                        append(Alpha1s,Rs,H2s),
                        append(H2s,Alpha2s,Betas).

lvong(Ws) :- sigma_stern(Ws), start(S), redrel_plus([S],Ws).

sigma_stern([ ]).
sigma_stern([A|Ws]) :- sigma(A),sigma_stern(Ws).

redrel_plus(Xs,Ys) :- redrel(Xs,Ys).
redrel_plus(Xs,Ys) :- redrel(Xs,Hs), redrel_plus(Hs,Ys).
```